

MATH 2270-1

Symmetric matrices, conics and quadrics

December 2, 2005

Since this is formally your third Maple project, although it is really just part of your homework, you may again work in groups of up to three individuals, and you may do this group work for the entire homework assignment due December 9. So, each group hands in one assignment, with the group member names prominently displayed.

Conic sections:

```
[ > restart:with(linalg):with(plots):#for computations and pictures
  with(student):#to do algebra computations like completing the
  square
```

Let's do the problem we did in class on Wednesday, using Maple. First, recall what we did by hand: We started with the equation

$$2x^2 + 2y^2 + 5xy = 1.$$

We rewrote this equation using a 2 by 2 symmetric matrix:

$$\begin{bmatrix} x & y \end{bmatrix} \begin{bmatrix} 2 & \frac{5}{2} \\ \frac{5}{2} & 2 \end{bmatrix} \begin{bmatrix} x \\ y \end{bmatrix} = 1.$$

By hand, we computed the eigenvalues and eigenvectors of this matrix. Since we're currently using Maple, we can verify Wednesday's work:

```
[ > A:=matrix(2,2,[2,5/2,5/2,2]);
  A := \begin{bmatrix} 2 & \frac{5}{2} \\ \frac{5}{2} & 2 \end{bmatrix}
  > eigenvectors(A); #compute eigenvalues, algebraic multiplicities,
  #and bases for each eigenspace
  \left[ \frac{9}{2}, 1, \{[1, 1]\} \right], \left[ \frac{-1}{2}, 1, \{[-1, 1]\} \right]
```

As I can show you now on the blackboard, and/or show you again in class on Monday, eigenvectors to symmetric matrices with different eigenvalues are always perpendicular. So, for this matrix A, I can define an orthonormal eigenbasis, and I'll choose to make it positively oriented. These two orthonormal basis vectors are the columns of the change of basis matrix S:

$$S := \begin{bmatrix} \frac{\sqrt{2}}{2} & -\frac{\sqrt{2}}{2} \\ \frac{\sqrt{2}}{2} & \frac{\sqrt{2}}{2} \end{bmatrix}$$

Precisely, if $\begin{bmatrix} u \\ v \end{bmatrix}$ are the coordinates of a point with respect to our new basis, and if $\begin{bmatrix} x \\ y \end{bmatrix}$ are the coordinates with respect to our standard basis, then

$$\begin{bmatrix} x \\ y \end{bmatrix} = [S] \begin{bmatrix} u \\ v \end{bmatrix}.$$

Then, from our discussion of diagonalization we know that for the diagonal matrix of eigenvalues

$$\Lambda := \begin{bmatrix} \frac{9}{2} & 0 \\ 0 & -\frac{1}{2} \end{bmatrix},$$

$$\Lambda = [S]^T [A] [S].$$

Putting all of this together, points whose standard coordinates satisfy the original equation

$$\begin{bmatrix} x & y \end{bmatrix} [A] \begin{bmatrix} x \\ y \end{bmatrix} = 1$$

have coordinates in the new basis which satisfy the equation


$$\begin{bmatrix} u & v \end{bmatrix} [S]^T [A] [S] \begin{bmatrix} u \\ v \end{bmatrix} = 1,$$

$$\begin{bmatrix} u & v \end{bmatrix} [\Lambda] \begin{bmatrix} u \\ v \end{bmatrix} = 1,$$

$$\frac{9u^2}{2} - \frac{v^2}{2} = 1.$$

We recognize this as a hyperbola opening along the u -axis, i.e. along the axis of our first basis vector. In fact, here's a picture:

```
> implicitplot(2*x^2+2*y^2+5*x*y= 1,
               x=-3..3,y=-3..3, grid=[100,100],
               color='black');
```



There are three things to notice:

(A) Just knowing the eigenvalues of the matrix A allowed us to deduce that the curve had to be a rotated hyperbola! In general, the eigenvalue information of a quadratic equation in two variables will almost tell you enough to deduce which conic it describes. If you have a quadratic equation in three variables, the eigenvalue information will limit the quadric surface possibilities, but you will often need to do more algebra to deduce exactly which creature you've captured.

(B) The spectral theorem, which we will talk about on Monday, says that symmetric matrices always have orthonormal eigenbases, so the recipe we followed above in a particular case, will let us diagonalize any symmetric matrix (and the corresponding quadratic expression which is called a quadratic form), finding new coordinates in which these "quadratic forms" and equations become diagonal (have no cross terms). There are important applications of this procedure, beyond the work with conics and quartics which we play with in this Maple file and the homework.

(C) We could be taking a lot more advantage of the fact that we're using Maple, to automate this process of identifying conics and quadric surfaces.

Here's how to make Maple help you.

Let's do #30 page 423 of the Kolman text handout: We wish to identify the conic of points whose standard coordinates $[x,y]$ satisfy the equation

$$33\sqrt{2}x - 31\sqrt{2}y + 8x^2 - 16xy + 8y^2 + 70 = 0.$$

Notice there are linear terms so this is a rotated AND translated conic! We will write this equation in matrix form, as

$$\mathbf{x}^T \mathbf{A} \mathbf{x} + \mathbf{B} \mathbf{x} + C = 0$$

where the vector \mathbf{x} is the coordinate vector $\begin{bmatrix} x \\ y \end{bmatrix}$, \mathbf{A} is a symmetric two by two matrix, \mathbf{B} is a one by two

matrix, C is a scalar. Remember how to turn the quadratic expression into a matrix product, by using the exact coefficients of the perfect squares in the diagonal entries, and half of the cross-term coefficients in the off-diagonal entries:

```
[ > A:=matrix(2,2,[8,-8,-8,8]);
```

```

#matrix for quadratic form
B:=matrix(1,2,[33*sqrt(2),-31*sqrt(2)]);
#row matrix for linear term
C:=70; # constant term

```

$$A := \begin{bmatrix} 8 & -8 \\ -8 & 8 \end{bmatrix}$$

$$B := [33\sqrt{2} \quad -31\sqrt{2}]$$

$$C := 70$$

```

> fmat:=v->evalm(transpose(v)*A*v
+ B*v +C) ;
#this function takes a vector v and computes
# transpose(v)Av + Bv + C ,
#which will be a one by one matrix
f:=v->simplify(fmat(v)[1]); #this extracts the entry of the one
#by one matrix fmat(v)

```

$$fmat := v \rightarrow \text{evalm}(' \&*'(' \&*'(\text{transpose}(v), A), v) + ' \&*' (B, v) + C)$$

$$f := v \rightarrow \text{simplify}(fmat(v))$$

```

> fmat(vector([x,y])); #should be a one by one matrix
f(vector([x,y])); #should get its entry, which is what we want.
#should agree with our example

```

$$[33\sqrt{2}x - 31\sqrt{2}y + 70 + (8x - 8y)x + (-8x + 8y)y]$$

$$33\sqrt{2}x - 31\sqrt{2}y + 70 + 8x^2 - 16xy + 8y^2$$

```

> eqtn:=f([x,y])=0;
#should be our equation

```

$$eqtn := 33\sqrt{2}x - 31\sqrt{2}y + 70 + 8x^2 - 16xy + 8y^2 = 0$$

```


> implicitplot(eqtn,x=-5..5,y=-5..5, color=black,
grid=[100,100]); #what conic is this???? (In
#any case, you can chuckle at Maple's picture)

```

```

>

```



Now let's go about the change of variables:

```
[ > data:=eigenvecs(A);#get eigenvectors
      data := [16, 1, {[-1, 1]}, [0, 1, {[1, 1]}]]
```

You pick things out of the object above systematically, using indices to work through the nesting of brackets:

```
[ > data[1];#first piece of data
      [16, 1, {[-1, 1]}]
[ > data[1][1];#eigenvalue
      16
[ > data[1][2];#algebraic multiplicity
      1
[ > data[1][3];#basis for eigenspace
      {[-1, 1]}
[ > data[1][3][1];#actual eigenvector
      [-1, 1]
```

So that's how to extract the eigenvectors. Remember once more that if you have two eigenvectors from a symmetric matrix with different eigenvalues, they are automatically perpendicular, and we can make a rotated ortho-normal eigenbasis!

```
[ > v1:=data[1][3][1];#first eigenvector
  v2:=data[2][3][1];#second eigenvector
  w1:=v1/norm(v1,2);#normalized
  w2:=v2/norm(v2,2);#normalized
      v1 := [-1, 1]
      v2 := [1, 1]
      w1 :=  $\frac{v1\sqrt{2}}{2}$ 
      w2 :=  $\frac{v2\sqrt{2}}{2}$ 
```

```

> S:=augment(w1,w2):#our orthogonal matrix
  if det(S)<0 then S:=augment(w2,w1):
  fi:
evalm(S); #I put it the "if" clause to make the
#new basis vectors right-handed

$$\begin{bmatrix} \frac{\sqrt{2}}{2} & -\frac{\sqrt{2}}{2} \\ \frac{\sqrt{2}}{2} & \frac{\sqrt{2}}{2} \end{bmatrix}$$

> f(evalm(S&*[u,v]));#do the change of variables...
#here [u,v] are the new-basis coordinates of a point
#and S is the change of coordinates matrix from new basis
#to standard basis.

$$2u - 64v + 70 + 16v^2$$

> simplify(%)#simplify it!

$$2u - 64v + 70 + 16v^2$$

> completesquare(%,u); #complete the square in u

$$2u - 64v + 70 + 16v^2$$

> completesquare(%,v);

$$16(v-2)^2 + 6 + 2u$$

> Eqtn:=%=0;
#this is the new equation

$$Eqtn := 16(v-2)^2 + 6 + 2u = 0$$


```

This is a translated parabola! Could you have predicted this just from the eigenvalues of the matrix? What eigenvalue property would indicate the possibility of an ellipse?

Let's collect all of our commands into one template . (People who like to program could also turn this into a procedure, which would let you input an equation, and which would output the conic information and a plot.) You can see from where the colons and semicolons are that the output will be the eigenvalues, the transition matrix, a plot, and an equation just short of standard form. (You can improve plots by picking a denser grid size.)

```

> restart;with(linalg):with(plots):with(student):
Warning, the protected names norm and trace have been redefined and unprotected
Warning, the name changecoords has been redefined
>
A:=matrix(2,2,[8,-8,-8,8]);
B:=matrix(1,2,[33*sqrt(2),-31*sqrt(2)]);
C:=70;
fmat:=v->evalm(transpose(v)*A*v
               + B*v + C):
f:=v->simplify(fmat(v)[1]):
eigenvals(A); #show the eigenvalues
data:=eigenvectors(A):

```

```

v1:=data[1][3][1]:#first eigenvector
v2:=data[2][3][1]:#second eigenvector
w1:=v1/norm(v1,2):#normalized
w2:=v2/norm(v2,2):#normalized
P:=augment(w1,w2):#our orthogonal matrix,
#unless we want to switch columns
if det(P)<0 then
    P:=augment(w2,w1):
fi:
evalm(P);
implicitplot(f([x,y])=0,x=-5..5,y=-5..5,
    grid=[100,100],color='black');
#increase grid size for better picture, but too big
#takes too long
f(evalm(P&*[u,v])):#do the change of variables
simplify(%):#simplify it!
completesquare(%,u): #complete the square in u
completesquare(%,v):#and v
Eqtn:=%=0;

```

You can check your book problems about conic sections using such a template, in the case of two distinct eigenvalues. (If both eigenvalues are the same, then A is similar to a multiple of the identity matrix, so IS a multiple of the identity matrix, so you won't need any fancy computer help.)

Quadric Surfaces:

Bonus Question : worth 10 midterm points (for each group member!) Create a template or procedure which will diagonalize, graph, and put into standard form, the solution set to any quadratic equation in 3 variables - i.e. analogous to the template above which works for conic sections. You need only create a template or procedure which works when there are 3 distinct eigenvalues. (If you successfully tackle the case of higher geometric multiplicity you will need several conditionals to allow for the structure of the eigenvectors array. Also, in that case you will need to use the Maple Gram-Schmidt command or your own, to get an orthonormal basis of the offending eigenspace basis.

To make 3-d plots you can use `implicitplot3d`. If you make your grid too fine the plot will take a long time to make, so try the default grid at first and adjust if necessary. You might also have to adjust your limits in the plot to get a better picture. You can manipulate 3d plots with your mouse. There are a lot of interesting plot options which you write into the command or access from the plotting toolbar. One that helps me see things is to use the "boxed" axes option.


The 3d plotting routines have been known to crash Maple, so save your file often. Sometimes what really happens is that a dialog box gets opened behind one of your windows, where you can't see it, so that it will appear that Maple has frozen. If this happens, minimize your maple window (the black dot at the upper left corner), and then re-display it.

Here's a plot to play with, #1 on page 430 of Kolman You should be able to move it around to make it look like a one-sheeted hyperboloid.

```

> implicitplot3d(x^2 + y^2 + 2*z^2 - 2*x*y - 4*x*z - 4*y*z + 4*x = 8,
    x=-5..5,y=-5..5,z=-5..5, axes='boxed');

```



How is that picture related to this calculation:

```
[ > A:=matrix(3,3,[1,-1,-2,-1,1,-2,-2,-2,2]);
```

$$A := \begin{bmatrix} 1 & -1 & -2 \\ -1 & 1 & -2 \\ -2 & -2 & 2 \end{bmatrix}$$

```
[ > eigenvalues(A);
```

-2, 2, 4

```
[ >
```

Could you find a rotated basis in 3-space which would put this quadric surface into diagonal form?

```
[ > eigenvects(A);
```

[2, 1, {[-1, 1, 0]}], [4, 1, {[1, 1, -2]}], [-2, 1, {[1, 1, 1]}]

```
[ >
```